3D3-5

数学入試問題における数式処理システムの性能評価

Performance Evaluation of Symbolic Computation System on University Entrance Examination Problems

矢吹 太朗 *1
Taro Yabuki

*1千葉工業大学 社会システム科学部
Faculty of Social Systems Science, Chiba Institute of Technology

In trying to solve mathematical problems by artificial intelligence, the problems are transformed to first order logic expression and solved by mathematical algorithms such as quantifier elimination. In this paper, we transform mathematical problems of the University of Tokyo entrance examination in 1988 by hand, and try to solve them by mathematical algorithms. By measuring computation time and required storage capacity, and by checking knowledge introduced by us, effectiveness and validity of the method are discussed.

1. はじめに

コンピュータを用いて数学の問題を解く試みは，大きく 2 つに分けられる．

第 1 に，コンピュータを数学のツールとして活用しようという試みがある．Computer based math*1 のような，コンピュータを導入して，数学教育を変革しようという提唱もなされている．手で問題を解く際と，コンピュータを利用して問題を解く際とでは，その解法が大きく異なる場合があるが，そうであれば，数学教育にコンピュータを導入すると，学習を通じて獲得される能力も大きく異なるものになり得る．

第 2 に，プロジェクト「ロボットは東大に入れるか」のような，人工知能のグランドチャレンジとしての試みがある．大学入試の数学の問題を人工知能によって解く試みでは，問題文を一階述語論理式に変換し，限量記号消去法（quantifier elimination, QE）のような数式処理アルゴリズムを適用するというアプローチが採用されている [松崎 13]．自然言語で書かれた問題文をコンピュータで処理できる形に自動変換するのが難しく，変換さえできれば問題は解けると思われるかもしれないが，必ずしもそうではない．

以上の背景を踏まえ，大学入試の数学の問題を，コンピュータ（数式処理システム）で処理できるように人手で変換しながら解くことによって，手で解く場合とコンピュータを利用して解く場合の考え方の違いを確認したり，数式処理システムの適用可能性を検討したりすることを目指す．

2. 手法

本研究で用いるハードウェアとソフトウェア，取り組む問題と解答方針を確認する．

2.1 ハードウェアとソフトウェア

Raspberry Pi Model B（ARM11 ファミリ 700 MHz，主記憶 512MB）上で動作する *Mathematica* 10.0 for Linux ARM (32-bit) (November 19, 2013) を利用する．

問題に取り組む際に用いる *Mathematica* の主な関数は以下の通りである*2．引数には省略できるものもある．

`AbsoluteTime[]` 1900 年 1 月 1 日からの秒単位による経過時間を与える．

`MaxMemoryUsed[]` セッションのデータを保存するのに使われる最大のバイト数を与える．

`MaxValue[{`*f*`, `*cons*`}, {`*x*`, `*y*`, ...}, `*dom*`]` 制約条件 *cons* の下での *f* の最大値を与える．同様に，`MinValue[]` は最小値を与える．

`Reduce[`*expr*`, `*vars*`, `*dom*`]` 領域 *dom* で，*vars* について方程式あるいは不等式を解き，限定子を除去することで，命題 *expr* を簡約する．

`Refine[`*expr*`, `*assum*`]` *expr* 中のシンボルが仮定 *assum* を満足する明示的な数式で置換された際に得られるような *expr* の形を与える．

`Solve[`*expr*`, `*vars*`, `*doms*`]` 領域 *doms* で，方程式の系 *expr* の解を変数 *vars* について求めようとする．

2.2 取り組む問題と解答方針

1988 年の東京大学理系数学の入試問題全 6 題に取り組む（本番の試験時間は 150 分．配点は 440 点中の 120 点）．ある年に出題された全問題を扱うのは，それが大学入試数学のよいサンプルになると思われるからである．1988 年の問題を扱うのは，東大入試数学における有名な難問 2 題が含まれているからである．実際，文献 [大数 93] には，第 2 問と第 3 問についてそれぞれ，「結論を予想するのは容易ですが，きちんとした解答をつくるのは極めて困難な超難問です」，「方針は立てにくいし，計算も十分にめんどうな難問です」という記載があり，文献 [聖文新社 08] では，「高度な読解力と図形的センス，強靭な論理的思考力と粘り強い計算力」を要する問題の代表例として，この 2 題が紹介されている．

取り組む問題と解答方針を以下に示す．

1988 年 東京大学 理系 第 1 問

xy 平面上の一次変換 f が次の 3 条件とみたすとする．

(i) 点 $(1, 0)$ は f により第 4 象限の内部にうつる．

(ii) 点 $(0, 1)$ は f により第 2 象限の内部にうつる．

(iii) 点 $(1, 1)$ は f により第 1 象限の内部にうつる．

このとき f には逆変換が存在することを示せ．また，点 P の像 f(P) が第 1 象限の内部にあれば，点 P も第 1 象限の内部にあることを示せ．

連絡先: https://twitter.com/yabuki/

*1 https://www.computerbasedmath.org/

*2 http://reference.wolfram.com/language/

1. 仮定を述語として記述する．
2. 証明したい命題，設問の前半は「一次変換に対応する行列の行列式が 0 でないこと」を，後半は「点 P の像を (x, y) として，その逆写像が第 1 象限にあること」を記述する．
3. `Refine[]` を使って命題が真であることを確認する（本研究では，これが証明の代わりになると仮定している）．

問題文には明示されていない，以下の知識を利用する．

(a) 2 次元平面の一次変換は，2 行 2 列の行列で表現できる．
(b) 一次変換に逆変換が存在することと，対応する行列の行列式が 0 でないことは同等である．

この問題を解くためのコードを以下に示す．

```
A = {{a, c}, {b, d}};
assum = And[
   With[{p = A.{1, 0}}, And[p[[1]] > 0, p[[2]] < 0]],
   With[{p = A.{0, 1}}, And[p[[1]] < 0, p[[2]] > 0]],
   With[{p = A.{1, 1}}, And[p[[1]] > 0, p[[2]] > 0]]];

Refine[Det[A] != 0, assum] (* 出力：True *)

Refine[
 With[{p = Inverse[A].{x, y}}, And[p[[1]] > 0, p[[2]] > 0]],
 And[x > 0, y > 0, assum]] (* 出力：True *)
```

1988 年 東京大学 理系 第 2 問

空間内に平面 α がある．一辺の長さ 1 の正四面体 V の α 上への正射影を S とし，V がいろいろと位置を変えるときの S の最大値と最小値を求めよ．
ただし，空間の点 P を通って α に垂直な直線が α と交わる点を P の α 上への正射影といい，空間図形 V の各点の α 上への正射影全体のつくる α 上の図形を V の α 上への正射影という．

1. 一辺の長さ 1 の正四面体の頂点の座標を適当に決める．
2. 頂点を座標変換する．
3. 頂点の xy 平面上への正射影を求める．
4. 頂点の正射影をすべて含む最小の凸多角形の面積を求める．
5. `MaxValue[]` と `MinValue[]` を使って面積の最大値と最小値を求める．

問題文には明示されていない，以下の知識を利用する．

(a) 平面 α は xy 平面だとしてよい．
(b) 「いろいろと位置を変える」という操作，つまり 3 次元空間での移動は，x, y, z それぞれの軸の周りの回転と平行移動の合成で表現できる．
(c) 頂点の正射影だけを考えれば十分である．
(d) 2 次元ベクトル `A`, `B`, `C` で作られる三角形の面積は，`Abs[Det[B - A, C - A]]/2` である．
(e) 平面上の 4 点を含む最小の凸多角形の面積は，4 点から作られる 4 個の三角形の面積の和の 1/2 である．
(f) 三角関数を含む式の最大値や最小値は，$A = \cos a, B = \sin a$ のように三角関数を変換し，$A^2 + B^2 = 1$ のような条件を追加すると，`MaxValue[]` や `MinValue[]` で計算できるようになることがある [穴井 11]．

この問題を解くためののコードを以下に示す．

```
Needs["PolyhedronOperations`"]
area[{A_, B_, C_}] := Abs[Det[{B - A, C - A}]]/2;

v1 = PolyhedronData["Tetrahedron", "VertexCoordinates"];
v2 = Map[Function[{p},
    RotationMatrix[c, {0, 0, 1}].
     RotationMatrix[b, {0, 1, 0}].
     RotationMatrix[a, {1, 0, 0}].p], v1];
v3 = Map[Function[{p}, p + {x, y, z}], v2];
v4 = Map[Function[{p}, Take[p, 2]], v3];

s1 = FullSimplify[Total[Map[area, Subsets[v4, {3}]]]/2];
s2 = FullSimplify[s1 /.
   {Cos[a] -> A, Sin[a] -> B, Cos[b] -> P, Sin[b] -> Q},
  And[A^2 + B^2 == 1, P^2 + Q^2 == 1]];

MaxValue[{s2, And[A^2 + B^2 == 1, P^2 + Q^2 == 1]},
 {A, B, P, Q}] (* 出力：1/2 *)

MinValue[{s2, And[A^2 + B^2 == 1, P^2 + Q^2 == 1]},
 {A, B, P, Q}] (* 出力：1/(2*Sqrt[2]) *)
```

1988 年 東京大学 理系 第 3 問

C を $y = x^3 - x$, $-1 \leq x \leq 1$ で与えられる xy 平面上の図形とする．次の条件をみたす xy 平面上の点 P 全体の集合を図示せよ．
「C を平行移動した図形で，点 P を通り，かつもとの図形 C との共有点がただ 1 点であるようなものが，ちょうど 3 個存在する．」

1. C を x 方向に a，y 方向に b 平行移動することにする．
2. C を平行移動した図形と C の共有点がただ 1 点だという条件を記述し，`condition` とする．
3. 点 P の座標を (X, Y) とし，これが C を平行移動した図形上の点だということと，条件 `condition` をまとめて `eqn` とする．
4. 条件 `eqn` をみたすような (a, b) がちょうど 3 個存在するという条件を，限定子を使って記述する．
5. `Reduce[]` を使って限定子を消去し，その結果得られる (X, Y) の関係を図示する．

問題文には明示されていない，以下の知識を利用する．

(a) $y = f(x)$ を x 方向に a, y 方向に b 平行移動すると $y - b = f(x - a)$ になる．
(b) 3 次方程式はたかだか 3 個の解を持つ．

a, b をパラメータとして持つ x の方程式 $f(x) = 0$ がただ 1 つの解を持つという条件は，次のように記述できる（実行は不要）．

```
Reduce[Exists[x1, f[x1] == 0,
  ForAll[x2, f[x2] == 0, x1 == x2]], {a, b}]
```

これを応用すると，方針 1 から 3 は次のように実行できる．

```
condition = Simplify[Reduce[
   Exists[x1, And[-1 <= x1 <= 1, -1 <= x1 - a <= 1,
     x1^3 - x1 == (x1 - a)^3 - (x1 - a) + b],
    ForAll[x2, And[-1 <= x2 <= 1, -1 <= x2 - a <= 1,
      x2^3 - x2 == (x2 - a)^3 - (x2 - a) + b],
      x1 == x2]], {a, b}, Reals]]
(* 出力：a^3 == 4 (a + b) && (-2 <= a < 0 || 0 < a <= 2) *)

eqn = And[Y == (X - a)^3 - (X - a) + b,
   -1 <= X - a <= 1, condition];
```

条件 `eqn` は a についての 3 次方程式と見なせる（b は a に従属する）．題意をみたす図形がちょうど 3 個存在するためには，この 3 次方程式がちょうど 3 個の解を持てばよい．

X, Y をパラメータとして持つ a の方程式 $f(a) = 0$ が異なる解を持つという条件は，次のように記述できる（実行は不要）．

```
Reduce[Exists[{a1, a2},
  And[f[a1] == 0, f[a2] == 0, a1 != a2]], {X, Y}]
```

これを応用して方針 4 以降を実行する．

```
region = Reduce[Exists[{a1, a2, a3, b1, b2, b3},
   And[
    eqn /. {a -> a1, b -> b1},
    eqn /. {a -> a2, b -> b2},
    eqn /. {a -> a3, b -> b3},
    (a1 != a2), (a2 != a3), (a3 != a1)
    ]], {X, Y}, Reals];

RegionPlot[region, {X, -1, 1}, {Y, -1, 1},
 PlotPoints -> 50, MaxRecursion -> 8] (* 出力は図 1(a). *)
```

1988 年 東京大学 理系 第 4 問

> xy 平面上で原点から傾き a（$a>0$）で出発し折れ線状に動く点 P を考える．ただし，点 P の y 座標はつねに増加し，その値が整数になるごとに動く方向の傾きが s 倍（$s>0$）に変化するものとする．
> P の描く折れ線が直線 $x=b$（$b>0$）を横切るための a,b,s に関する条件を求めよ．

1. $y=n$ に対応する x を求める．
2. x が b より大きくなるような n が存在するということを，限定子を使って記述する．
3. `Reduce[]` を使って限定子を消去し，a, b, s に関する条件を求める．

問題文には明示されていない，以下の知識を利用する．

(a) y の値が整数の点のみを調べれば十分である．
(b) $\sum_{i=1}^{n} x^{i-1}$ は $x\neq 1$ なら $(x^n-1)/(x-1)$，$x=1$ なら n である（単純に `Sum[x^(i - 1), {i, 1, n}]` とすると，$x=1$ の場合が考慮されない）．

この問題を解くためののコードを以下に示す．

```
Simplify[Reduce[Exists[n, Or[
    And[s != 1, b < (s^(1 - n) (s^n - 1))/(a (s - 1))],
    And[s == 1, b < n/a]]],
  {a, s, b}, Reals], And[0 < a, 0 < b, 0 < s]]
(* 出力：s <= 1 || b < s/(a*(-1 + s)) *)
```

1988 年 東京大学 理系 第 5 問

> xyz 空間において，xz 平面上の $0\le z\le 2-x^2$ で表される図形を z 軸のまわりに回転して得られる不透明な立体を V とする．V の表面上 z 座標 1 のところにひとつの点光源 P がある．xy 平面上の原点を中心とする円 C の，P からの光が当たっている部分の長さが 2π であるとき，C のかげの部分の長さを求めよ．

1. 点 P の座標を $(1,0,1)$ とする．
2. 点 P を通る V の接平面を求める．
3. 接平面と xy 平面の交線 $x=X$ を求める．
4. 光が当たる部分の弧に対応する中心角を t として，方程式 $2rt=2\pi, r\cos t=X$ を立てる．
5. `Solve[]` を使って方程式を解き，r を求める．
6. 求める長さ，$2\pi r-2\pi$ を計算する．

問題文には明示されていない，以下の知識を利用する．

(a) 対称性から，点 P の座標は $(1,0,1)$ としてよい．
(b) かげになるのは，点 P における V の接平面に対して，V と同じ側にある領域である．
(c) 接平面は $a(x-1)+by+(z-1)=0$ の形で記述できる．

この問題を解くためののコードを以下に示す．

```
cond = Reduce[ForAll[{x, y, z},
    And[z == 2 - (x^2 + y^2),
     a (x - 1) + b y + (z - 1) == 0],
    And[x == 1, y == 0, z == 1]], {a, b}, Reals];
splane = (a (x - 1) + b y + (z - 1) == 0) /.
   First[Solve[cond, {a, b}]];
X = x /. First[Solve[splane /. {z -> 0}, x]];
sol = Solve[And[2 r t == 2 Pi, r Cos[t] == X], {r, t}, Reals];
(2 Pi r - 2 Pi) /. First[sol] (* 出力：4 Pi *)
```

1988 年 東京大学 理系 第 6 問

> 空間内の点 O に対して，4 点 A, B, C, D を
>
> $$\mathrm{OA}=1,\ \mathrm{OB}=\mathrm{OC}=\mathrm{OD}=4$$
>
> をみたすようにとるとき，四面体 ABCD の体積の最大値を求めよ．

1. A, B, C, D の座標をパラメータ表示する．
2. 四面体 ABCD の体積を求める．
3. `MaxValue[]` を使って体積の最大値を求める．

問題文には明示されていない，以下の知識を利用する．

(a) A は $(0,0,1)$ としてよい．
(b) A が $(0,0,1)$ なら，四面体 ABCD の体積が最大になるのは，BCD を含む平面が xy 平面に平行なときである．
(c) 3 次元ベクトル `a`, `b`, `c`, `d` で作られる四面体の体積は，`Cross[b - a, c - a].(d - a)/6` である．

この問題を解くためののコードを以下に示す．

```
a = {0, 0, 1};
b = {4 Cos[t], 0, -4 Sin[t]};
c = {4 Cos[t] Cos[u], 4 Cos[t] Sin[u], -4 Sin[t]};
d = {4 Cos[t] Cos[v], 4 Cos[t] Sin[v], -4 Sin[t]};
f = Simplify[Cross[b - a, c - a].(d - a)/6]
```

体積 f は，次のように t の関数と u と v の関数の積になっている．

$$f=\frac{32}{3}(4\sin t+1)\cos^2 t\sin\left(\frac{u}{2}\right)\sin\left(\frac{v}{2}\right)\sin\left(\frac{u-v}{2}\right).$$

このため，f の最大値は次のように求められる．

```
part1 = 32 Cos[t]^2 (1 + 4 Sin[t]);
part2 = f/part1;

With[{
  p1min = MinValue[{part1, 0 <= t < Pi}, {t}],
  p1max = MaxValue[{part1, 0 <= t < Pi}, {t}],
  p2min = MinValue[{part2,
     And[0 <= u < 2 Pi, 0 <= v < 2 Pi]}, {u, v}],
  p2max = MaxValue[{part2,
     And[0 <= u < 2 Pi, 0 <= v < 2 Pi]}, {u, v}]},
 If[And[p1max > 0, p2max > 0],
  If[And[p1min < 0, p2min < 0],
   Max[p1max p2max, p1min p2min],
   p1max p2max]]] (* 出力：9 Sqrt[3] *)
```

3. 結果

全 6 問の解答の結論（第 3 問の結論は図 1(a)，他の問題の結論はコード中に記載）は，[大数 93] に掲載されているものと一致していた．

計算時間と利用記憶容量は表 1の通りである．



(a) 本手法の結果



(b) 模範解答

図 1: 第 3 問の解答

表 1: 計算時間と利用記憶容量

問題	計算時間（秒）	利用記憶領域（MB）
第 1 問	4.6	10
第 2 問	3.3×10^3	35
第 3 問	1.5×10^2	71
第 4 問	3.2×10	18
第 5 問	6.8×10	21
第 6 問	7.0×10	17

4. 考察

大学入試の数学の問題を数式処理システムを使って解くためには，問題文をそのまま数式処理システム用の言語に置き換えるのではなく，問題についての理解に基づいて対称性などを発見し，計算量を減らす工夫が必要である．

数学の学習に数式処理システムを導入しようとする場合には，手で解く場合と，数式処理システムを利用して解く場合とで，解き方が大きく異なることに注意しなければならない．数式処理システムを利用した学習によって身につく能力と，数学を学習して身につけることが期待されてきた能力の違いを明らかにするような研究が必要であろう．

結論 (真偽値や数値, 図形) が模範解答のそれと一致しているからといって，ここで得られた結果が数学的に正しく，大学入試で正解と見なされるというわけではない．たとえば，*Mathematica* 7 の `Reduce[]` は，およそ 350 ページの *Mathematica* コードとおよそ 1400 ページの C コードが使われているというが [Wolfram 11]，試験の採点者にそのコードを検証させるのは現実的ではないだろう．

4.1 各論

4.1.1 第 1 問

後半の問いに対して，点 P 自体を (x, y) として解くことはできなかったため，点 P の像を (x, y) として命題を記述した．変数の導入方法には，試行錯誤が必要である．

4.1.2 第 2 問

知識 (e) を受験数学で学ぶことはないと思われるが，`MaxValue[]` や `MinValue[]` を利用するためには，このような，プログラムではなく数式で表現するための知識が有用であろう．

知識 (f) を用いずに，面積を三角関数で記述したままでは，`MaxValue[]` や `MinValue[]` で最大値や最小値を求めることはできなかった．三角関数を含む式に関する問題では，このような変数変換が有力なテクニックである．

難問として有名な本問だが，数式処理システムを利用すれば，比較的簡単に解ける．しかしその解法は，文献 [小島 89] などに掲載されているような，人間が手で解く方法とは大きく異なっている．

4.1.3 第 3 問

a についての 3 次方程式になることがわかって初めて知識 (b) が利用できる．このように，具体的な方針を事前に決めるのが難しい場合がある．

本手法における第 3 問の結論は図 1(a) だが，模範解答は図 1(b) のように，曲線の式や交点の座標，曲線自体や点自体を含むかどうかも描いたものになる．

4.1.4 第 4 問

Mathematica では，$\sum_{i=1}^{n} x^{i-1}$ のような簡単な計算が，特殊な仮定（この場合は $x \neq 1$）の下に行われる危険があることを知っておかなければならない．

`Reduce[]` の引数の {`a`, `s`, `b`} を {`a`, `b`, `s`} にすると解けない．変数の順番が大切である．

4.1.5 第 5 問

接平面を $(x-1) + ay + b(z-1) = 0$ の形で記述すると，計算時間は 68 秒から 3300 秒に，利用記憶容量は 21 MB から 230 MB に増加する．このように，得られる結果は同等でも，変数の導入方法によって計算時間は大きく異なる場合がある．

4.1.6 第 6 問

問題をそのまま定式化しても，現実的な時間では解けない．(a)，(b) のような知識によって，計算量を減らす必要がある．

体積 f を，t の関数と u と v の関数に分けずに，`MaxValue[]` で最大値を求めることはできなかった．f を 2 つの部分に分けられることは，解いている途中で初めてわかることである．第 3 問と同様，具体的な方針を事前に決めるのは難しい．

5. おわりに

大学入試の数学の問題で，数式処理システムを用いて現実的に解けるものを確認した．その解法は手で解く場合のそれとは異なり，問題についての理解に基づく知識も必要なものであった．このような事例を積み重ねることによって，数学教育が改善したり，人工知能研究が進展することが期待される．

参考文献

[Wolfram 11] Wolfram Research, I.: Some Notes on Internal Implementation (2011)

[穴井 11] 穴井 宏和, 横山 和弘：QE の計算アルゴリズムとその応用, 東京大学出版会 (2011)

[小島 89] 小島 寛之：ルシャリーに花束を，大学への数学, Vol. 33, No. 11–12, pp. 56–57 (1989)

[松崎 13] 松崎 拓也, 岩根 秀直, 穴井 宏和, 相澤 彰子, 新井 紀子：深い言語理解と数式処理の接合による入試数学問題解答システム, 人工知能学会全国大会論文集 (CD-ROM), pp. 2A4–1 (2013)

[聖文新社 08] 聖文新社 編集部（編）：東京大学 数学入試問題 50 年, 聖文新社 (2008)

[大数 93] 大学への数学 編集部（編）：東大入試・10 年の軌跡, 東京出版 (1993)